# Laser Cooling

When we load ions into this trap they are confined along the axis of the trap but, as shown in the previous notebook, they can execute harmonic motion in the xy plane. Our goal is to align ions along the axis and minimize xy motion as much as possible. From the previous example we see that we can do this by making $v_{ox}$, $v_{oy}$ as small as possible. Since velocity can be related to kinetic energy, which in turn, can be characterized by a temperature, the process of minimizing the kinetic energy of the trapped ion is called cooling. A powerfull method for accomplishing this is called laser cooling.

Atoms and ions have internal structure, they are called energy levels and in an oversimplistic description we can consider our ions to have two energy levels as shown in the picture below




Roughly speaking, the ion absorbs a photon with energy $\hbar\omega$ that is equal to the difference in the energies of the two levels of the ion. Here $\hbar$ is Plank's constant and $\omega$ is the angular frequency of the photon which is related to its wavelength $\lambda$ by the relation $\omega = 2\,\pi\,c/\lambda$ where c is the speed of light. In addition to absorbing the energy of the photon the ion can also "gain" a momentum kick from that photon. This momentum is a vector and is given by the expression

$$\hbar \boldsymbol{k}$$

where $\boldsymbol{k}$ is a vector that points in the propagation direction of the photon and has the magnitude $|\boldsymbol{k}| = \frac{\lambda}{c}$. The frequency $\omega_0$ that satisfies the identity

$$\Delta E \equiv E_e - E_g = \hbar\omega_0$$

is called the resonance frequency of that particular transition. If we shine light near that frequency on the atom, in its ground state, it excites into states near $|e\rangle$ . Almost immediately ( in $10^{-6} - 10^{-9}$ seconds, depending on the system) it will then de-excite and emit photons in random directions. This is due to a process called spontaneous emission (also called fluorescence). The formula that gives us the number of photons, of given frequency $\omega$, emitted per unit time is

$$N(\omega) = N(\omega_0)\frac{(\Gamma/2)^2}{(\omega-\omega_0)^2 + (\Gamma/2)^2}$$

```
Clear[ω]
parameters = {n0 → 10, ω0 → 3, Γ → 0.2, ℏ → 1, c → 10, mass → 1};

nphotons[ω_] = n0 (Γ / 2) ^2 / ((ω - ω0) ^2 + (Γ / 2) ^2)

Plot[nphotons[ω] /. parameters, {ω, 0, 6}, PlotRange → All]
```

Note that the function N($\omega$) is highly peaked near the resonance frequency $\omega_0$ and the width of the curve is determined by the
parameter Γ which is also called the full width at half maximum (FWHM). This type of function is called a Lorentzian
and these parameters depend on which ion one chooses.Though N($\omega$) determines the fluorescence spectrum, it also determines the efficiency in which a photon of frequency $\omega$ can be absorbed by the ion. The closer $\omega$ is to the resonance frequency the more likely the ion will absorb the photon.

Notice also that the integral $\int N(\omega)\, d\omega$ = N($\omega_0$) $\frac{\pi\,\Gamma}{2}$.

```
Integrate[(Γ / 2) ^2 / ((ω - ω0) ^2 + (Γ / 2) ^2), {ω, -∞, ∞}, Assumptions → ω0 > 0 && Γ > 0]
```

$$\frac{\pi\,\Gamma}{2}$$

If an atom absorbs a photon of momentum $\boldsymbol{k}$ in an interval of time $\Delta t$ it experiences a force

$$\Delta F \equiv \boldsymbol{F}_{rad} = \frac{\hbar \boldsymbol{k}}{\Delta t}$$

This is called radiation pressure force and therefore

$$\boldsymbol{F}_{rad} = N(\omega)\hbar \boldsymbol{k}$$

Thus the ion will see a net force in the direction determined by $\boldsymbol{k}$. However the ion also emits photons, i.e. it fluoresces! Each time the ion emits a fluorescence photon the ion must recoil in the opposite direction in order to conserve momentum. After several photons emissions, each in an arbitrary direction, the total recoil momentum averages to zero, though the ion does have any average recoil energy. Therefore at the end of the day the ion will experience a net radiation pressure force given by the expression above.

Radiation pressure



For a transparent narrative let's consider the following situation in 1D. An ion is trapped near the origin, x=0, in a harmonic well with potential

$$V = \frac{1}{2} m \Omega^2 x^2$$

A laser beam shines light arriving from the left with frequency $\omega_L$. Since the ion is moving at velocity $v$ with respect to our rest frame it sees, according to the Doppler shift, light of frequency

$$\omega' = \omega_L \sqrt{\frac{1 - v/c}{1 + v/c}}$$

Thus the radiation pressure force is

$$F_{rad} = \frac{\hbar \omega'}{c} N(\omega')$$

in a direction along the x-axis. Now in general $v/c << 1$ and we can make a Taylor expansion to get (see below )

$$F_{rad} \approx \frac{\hbar \omega_L}{c} N(\omega_L) - \frac{\hbar \omega_L^2}{c^2} N'(\omega_L) v$$

where $N'(\omega_L)$ is the derivative of $N(\omega_L)$. Now if $\omega_L < \omega_0$ then $N'(\omega_L) > 0$ and the velocity dependent form has a negative sign. If we now equate to total force on our particle is a sum of the harmonic force $-m\Omega^2 x$ and the radiation pressure force given above, we have

$$m\ddot{x} = -m\Omega^2 x + F_0 - \beta\dot{x}$$
$$F_0 = \frac{\hbar\omega_L}{c} N(\omega_L)$$
$$\beta = \frac{\hbar\omega_L^2}{c^2} N'(\omega_L)$$

This differential equation can be solved and we get the analytic solution

$$x(t) = x_s + x_0 \exp(-\beta t/2) \cos(\tilde{\Omega}\, t) + \frac{p_0}{m\Omega} \exp(-\beta t/2) \sin(\tilde{\Omega}\, t)$$

where

$$x_s = \frac{F_0}{m\Omega^2}$$
$$\tilde{\Omega} = \sqrt{\Omega^2 - (\beta/2)^2}$$

and $x_0, p_0$ are constants.

```
(* with the Doppler shifted frequency the radiation force becomes *)
ℏ ω Sqrt[1 - v / c] / Sqrt[1 + v / c] / c nphotons[ω Sqrt[1 - v / c] / Sqrt[1 + v / c]]
ftotal[ω_] = % /. v / c → γ

Normal[Series[ftotal[ω], {γ, 0, 1}]]

(* f0 is velocity independent *)
f0[ω_, v_] = FullSimplify[n0 Γ^2 ω ℏ / (c (Γ^2 + 4 ω^2 - 8 ω ω0 + 4 ω0^2))]
(* f1 depends linearly on the velocity *)
f1[ω_, v_] = FullSimplify[-(n0 γ Γ^2 ω (Γ^2 - 4 ω^2 + 4 ω0^2) ℏ) / (c (Γ^2 + 4 ω^2 - 8 ω ω0 + 4 ω0^2)^2)] /. γ → v / c
```

$$\frac{n0\,\Gamma^2\,\omega\,\hbar}{c\,\left(\Gamma^2 + 4\,(\omega - \omega0)^2\right)}$$

$$-\frac{n0\,v\,\Gamma^2\,\omega\,\left(\Gamma^2 - 4\,\omega^2 + 4\,\omega0^2\right)\,\hbar}{c^2\,\left(\Gamma^2 + 4\,(\omega - \omega0)^2\right)^2}$$

```
Plot[f0[ω, 0] /. parameters, {ω, 0, 4}, PlotRange → All]
Plot3D[(f0[ω, v] /. parameters),
 {ω, 0, 4}, {v, -1, 1}, PlotRange → All, PlotPoints → 50]
```

```
Plot3D[f1[ω, v] /. parameters, {ω, 0, 4}, {v, -1, 1}, PlotRange → All, PlotPoints → 50]

frad[ω_, t_] := ℏ ω / c Sqrt[1 - x'[t] / c] / Sqrt[1 + x'[t] / c]
   nphotons[ω Sqrt[1 - x'[t] / c] / Sqrt[1 + x'[t] / c]] / mass /. parameters

Omega = 3;
omega = ω0 - Γ /. parameters
x0 = 1;
v0 = 1;

sols =
 Flatten[NDSolve[ {x''[t] == - Omega^2 x[t] + frad[omega, t], x[0] == x0, x'[0] == v0},
   x[t], {t, 0, 100}]]
fun[t_] = x[t] /. sols

Plot[fun[t], {t, 0, 100}, PlotRange → All]

F0 = ℏ omega / c nphotons[omega] /. parameters;
Print[" F0=", F0]
beta = ℏ^2 omega^2 / c^2 nphotons'[omega] /. parameters;
Print["β=", beta]
xs = F0 / mass / Omega^2 /. parameters;
Print["xs=", xs]
Omegatilde = Sqrt[Omega^2 - (beta / 2)^2];
Print["Omegatilde=", Omegatilde]

fun2[t_] = xs + x0 Exp[-beta t / 2] Cos[Omegatilde t] +
   v0 / Omega Exp[-beta t / 2] Sin[Omegatilde t];

Plot[fun2[t], {t, 0, 100}, PlotRange → All]
```